



# DECUS

## PROGRAM LIBRARY

DECUS NO.	8-279
TITLE	BAR CHART PLOTTING SUBROUTINE
AUTHOR	G. L. Kermez and W. G. Peters
COMPANY	Texas Instruments Limited Bedford, England
DATE	January 27, 1970
SOURCE LANGUAGE	MACRO-8



## BAR CHART PLOTTING SUBROUTINE

DECUS Program Library Write-up

DECUS No. 8-279

### INTRODUCTION

The program was written as part of the main operating program of a power transistor test system produced at T.I.L., Bedford by the authors.

### METHOD OF OPERATION

The chart requires 7 constants for its construction, i.e.,

- (1) the octal location of the table which will be charted;
- (2) the octal number of readings in table;
- (3) the octal number of the table increment in which the reading required occurs, i.e., 3000, 1000, 3 would imply, read the table starting in location 3000, and chart every third reading until 1000 readings have been read;
- (4) the octal number of classes into which the readings in the table are to be distributed;
- (5) the octal value of each increment of the Y axis; there are 50 increments normally displayed across the T.T.Y. paper;
- (6) the octal value of the lower limit of the total class range; and
- (7) the octal value of the upper limit of the total class range. The subroutine itself calculates the individual class span. Each class is inclusive of the lower limit, but exclusive of the upper limit.

The table can be as large as required. Many charts can be plotted from one table, because the table is not altered in any way by the subroutine. This allows the drawing of a family of charts of one set of readings for different classes and numbers.


The subroutine contains interger divide, and decimal print routines. Both of these can be used by other parts of a program. The normal ASR33 service routines, i.e., JMS TYPE and JMS CARLF are required. The CCONST subroutine allows complete freedom to change the constants set-up method. It must be stressed that because of the simplicity of the interger divide subroutine employed, errors, when dealing with large numbers, may occur. However, these may be minimised by intelligent manipulation of the axis values.

The following examples illustrate the flexibility of the subroutine. They show a range of charts of instructions in a program, with varying axis values.



FIGURE 1

C,1,10,10,0,4000,



BAR CHART OF 2000 INSTRUCTIONS  
0 to 4000, IN 10 CLASSES

FIGURE 2

C,1,10,5,0,400,

```

0    25    50    75    100   125   150   175   200   225   250
I---I---I---I---I---I---I---I---I---I---I---I---
0
I*****
40
I**
80
I**
120
I****
160
I*****
200
I*****
240
I***
280
I*
320
I*****
360
I***
400
```

SAME INSTRUCTIONS AS FIGURE 1, TOTAL  
CLASS CONSIDERED NOW 0 - 400

FIGURE 3

C,1,5,4,0,40,

```

0    20    40    60    80    100   120   140   160   180   200
I---I---I---I---I---I---I---I---I---I---I---I---
0
I*****
8
I*****
16
I*
24
I
32
I**
40
```

SAME AS FIGURE 1  
TOTAL CLASS NOW 0 - 40

FIGURE 4

C,1,20,2,2400,2800,

	0	10	20	30	40	50	60	70	80	90	100	
	I	---	I	---	I	---	I	---	I	---	I	---
2400	I	*****										
2420	I	*										
2440	I											
2460	I											
2480	I											
2500	I											
2520	I	*****										
2540	I	*****										
2560	I											
2580	I											
2600	I											
2620	I	*										
2640	I											
2660	I											
2680	I	*****										
2700	I	*****										
2720	I	*****										
2740	I	****										
2760	I	*****										
2780	I	*****										
2800												

SAME INSTRUCTIONS AS FIGURE 1.  
 CLASS CONSIDERED IS 2400 - 2800  
 IN 20 INCREMENTS



```

L
001 / FOLLOWS PROGRAM TO PRINT OUT BAR CHART OF GIVEN
002 / LIST OF VALUES.
003 / IT IS WRITEN AS A SUBROUTINE TO EASE CALL-UP
004 *1000/ RELOCATABLE ORIGIN
005 CHARTS,0
006 JMS CCONST/ SET UP CONSTANTS SUBROUTINE
007 JMS CARLF;JMS CARLF/ GET SOME PAPER!!
008 JMS YLINE/ PRINTOUT OF Y AXIS AND APRO. CONSTANTS
009 JMS COMPARE/ COMPARISON OF TABLE WITH CLASS INCREMENTS
010 TAD SAVE4 /GET UPPER LIMIT OF FINAL CLASS
011 JMS DECPRT /PRINT IT
012 JMS CARLF;JMS CARLF/ GET MORE PAPER!!
013 JMP I CHARTS/ EXIT
014 /
015 /FOLLOWS SUBROUTINE TO PRINT OUT YAXIS AND THE
016 / APROPRIATE GRADINGS
017 /
018 YLINE,0
019 TAD (-13
020 DCA CNT1/ INITIALISE POINTER
021 DCA CNT2/ ZERO CNT2
022 YONWARD,CLA IAC/ SET ACC TO 1
023 DCA SAVEC
024 JMS SPACES/ PRINT 2 SPACES
025 TAD CNT2
026 JMS DECPRT/ PRINT OUT 0
027 ISZ CNT1
028 JMP .+2
029 JMP YAXIS/ EXIT WHEN GRADINGS ALL PRINTED
030 TAD CNT2
031 TAD SAVE5/ CONTAINING GRADING INCREMENT
032 DCA CNT2 / UPDATE CNT2
033 JMP YONWARD/ CARRY ON PRINTING GRADINGS
034 YAXIS,JMS CARLF/ PRINTING OF GRATICULE
035 TAD (-13
036 DCA CNT1/ RE INITIALISATION OF CNT1
037 TAD (-4
038 DCA SAVEC
039 JMS SPACES/ PRINT 4 SPACES
040 YGRAT,TAD (311/ LETTER I
041 JMS TYPE / AND TYPE IT
042 TAD (-4
043 DCA CNT3 / POINTER FOR 4 YGRATICULE DIVISIONS
044 TAD (255/ LETTER-
045 JMS TYPE
046 ISZ CNT3
047 JMP .-3
048 ISZ CNT1
049 JMP YGRAT/ CARRY ON PRINTING GRATICULE
050 JMS CARLF/ PRINTING DONE
051 JMP I YLINE
052 CNT1,0
053 CNT2,0
054 CNT3,0
055 SAVE5,0
056 PAUSE

```

```

L
001 / FOLLOWS SUBROUTINE TO RUN THRO A TABLE
002 /AND COUNT NUMBER OF TIMES A READING
003 /FALLS WITHIN A GIVEN CLASS
004 /
005 COMPARE,0
006 EVER,TAD PUSH/ GET START OF TABLE ADDRESS
007 DCA SAVEG / TEMP. STORE
008 DCA SAVEM
009 TAD SAVEK/ GET NUMBER OF READINGS IN TABLE
010 CIA
011 DCA SAVEE/ TEMP STORE
012 ONWARD,TAD I SAVEG/ GET FIRST READING FROM TABLE
013 CIA
014 DCA SAVE1
015 TAD SAVE1
016 TAD SAVEA/ TAD MIN VALUE OF CLASS
017 SMA SZA CLA/ SKIP IF RESULT >= LOWER
018 JMP MANDY / RESULT IS LESS THAN LOWEST CLASS
019 TAD SAVE1
020 TAD SAVE4/ TAD MAX VAL UE OF CLASS
021 SPA SNA CLA/ SKIP IF RESULT < UPPER
022 JMP MANDY/ RESULT IS MORE THAN UPPER CLASS BOUNDARY
023 ISZ SAVEM/ INCREMENT CLASS COUNTER
024 NOP
025 MANDY,TAD SAVEG/ GET START OF TABLE ADDRESS
026 TAD SAVEF/ ADD INCREMENT
027 DCA SAVEG/ START OF TABLE POINTER UPDATED
028 ISZ SAVEE/ TABLE DONE????
029 JMP ONWARD/ NO... GET NEXT RESULT IN TABLE
030 TAD SAVEA/ YES! UPDATE CLASS LIMITS
031 JMS DECPRT/ PRINT LOWER LIMIT OF CLASS
032 JMS CARLF
033 JMS ASTPRT/ PRINT OUT NUMBER OF*
034 JMS CARLF
035 TAD SAVEA
036 TAD SAVE2 / ADD CLASS INCREMENT
037 DCA SAVEA/LOWER LIMIT INCREMENTED
038 ISZ SAVED/ ANY MORE CLASSES????
039 JMP .+2 / YES
040 JMP I COMPARE/ NO. EXIT SUBROUTINE
041 TAD SAVE4/ GET MAX VALUE OF CLASS
042 TAD SAVE2/ TAD INCREMENTAL INCREASE
043 DCA SAVE4/ NEW MAX VALUE
044 JMP EVER / CARRY ON WITH COMPARISONS
045 SAVEG,0
046 SAVEM,0
047 SAVEE,0
048 SAVE1,0
049 SAVE4,0
050 PUSH,0
051 SAVEK,0
052 SAVEA,0
053 SAVEF,0
054 SAVE2,0
055 SAVED,0
056 PAUSE

```



```

L
001 / FOLLOWS SUBROUTINE TO PRINT NUMBER OF SPACES
002 /GIVEN IN SAVEC
003 /
004 SPACES,0
005 CLA
006 TAD SAVEC
007 CIA
008 DCA CNT3/ INITIALISE
009 TAD (240
010 JMS TYPE
011 ISZ CNT3
012 JMP .-3
013 JMP I SPACES
014 SAVEC,0
015 PAGE
016 /FOLLOWS SUBROUTINE TO PRINT OUT A DEFINED
017 / NUMBER OF "*"
018 /
019 /
020 ASTPRT,0
021 CLA CLL
022 TAD SAVEM/GET NUMBER IN CLASS
023 DCA ADIVI/ SET NUMERATOR IN DIVIDE ROUTINE
024 TAD SAVEB/ GET VERTICAL SCALING FACTOR
025 JMS DIVIDE
026 SZA/ SKIP IF NO "*" REQUIRED
027 JMP .+2
028 JMP ZEROAST
029 CIA
030 DCA ASTSAVE/ SET UP * COUNTER
031 TAD (4
032 DCA SAVEC
033 JMS SPACES/ TYPE 4 SPACES
034 TAD (311 /LETTER I
035 JMS TYPE
036 /
037 ASTTYPE,TAD (252/ LETER *
038 JMS TYPE
039 ISZ ASTSAVE/ANY MORE?
040 JMP ASTTYPE/ YES CARRY ON TYPING *
041 JMP I ASTPRT/ NO, EXIT
042 /
043 ZEROAST,TAD (4/ NO * REQUIRED
044 DCA SAVEC
045 JMS SPACES/ PRINT 4 SPACES
046 TAD (311/ LETTER I
047 JMS TYPE/ PRINT IT
048 JMP I ASTPRT/ EXIT
049 ASTSAVE,0
050 SAVEB,0
051 PAUSE
*

```

```

L
001 / FOLLOWS INTERGER DIVIDE SUBROUTINE
002 /
003 DIVIDE,0
004 DCA BDIVI/ STORE OF DENOMINATOR
005 DCA CDIVI/ ZERO QUOTIENT
006 TAD BDIVI
007 SNA / DIVIDE BY ZERO EXIT
008 JMP I DIVIDE
009 CIA
010 DCA BDIVI
011 CLL
012 TAD ADIVI
013 DIVLOP,TAD BDIVI/ SUBTRACT DENOM.
014 SNA/ DIVISION EXACT INCREMENT QUO, EXIT
015 JMP EXACT
016 SZL
017 JMP CARYON / DIVISION INCOMPLETE
018 CLA CLL
019 SKP
020 EXACT,ISZ CDIVI / INCREMENT QUOTIENT
021 TAD CDIVI
022 JMP I DIVIDE/ EXIT WITH ANS IN ACC
023 CARYON,ISZ CDIVI
024 CLL
025 JMP DIVLOP/ CONTINUE SUBTRACTING DENOM.
026 BDIVI,0
027 CDIVI,0
028 ADIVI,0
029 /
030 /
031 PAUSE

```

\*

```

L
001 / FOLLOWS CONSTANTS SET-UP SUBROUTINE
002 / IT IS WRITEN SO TO ALLOW EASY CHANGES
003 / FOR INCORPORATION INTO EXISTING PROGRAMS
004 / IN THE ORIGINAL, ALL THE SAVE LOCATIONS
005 / WERE IN PAGE 0, ACTING AS A SCRATCH STORE.
006 /
007 CCONST,0
008 CLA CLL
009 TAD START/GET START ADDRESS OF TABLE
010 DCA PUSH
011 TAD NUMBER/ GET NO. OF READINGS IN TABLE
012 DCA SAVEK
013 TAD INCRM/ GET TABLE SEARCH INCREMENT
014 DCA SAVEF
015 TAD NCLASS/ GET NUMBER OF CLASSES
016 CIA
017 DCA SAVED
018 TAD VSF / GET VERTICAL SCALING FACTOR
019 DCA SAVEB
020 TAD (-5/ FOLLOWS X5 MULT.
021 DCA TEMP
022 TAD VSF
023 ISZ TEMP
024 JMP .-2
025 DCA SAVE5/ THIS IS VSF X5
026 TAD LOLIM/ GET LOWER LIMIT
027 DCA SAVEA
028 TAD LOLIM/ FOLLOS CALC TO GET CLASS SEPARATION
029 CMA IAC
030 DCA TEMP
031 TAD UPLIM
032 TAD TEMP
033 DCA ADIVI
034 TAD NCLASS
035 JMS DIVIDE
036 DCA SAVE2/ CLASS SEPARATION
037 TAD SAVE2
038 TAD LOLIM
039 DCA SAVE4
040 JMP I CCONST/ EXIT
041 / FOLLOWS CHART BUILD-UP CONSTANTS
042 /
043 START,0/ START OF TABLE ADDRESS
044 NUMBER,0/ NUMBER OF READINGS IN TABLE
045 INCRM,0/TABLE INCREMENT FACTOR
046 NCLASS,0/ NUMBER OF CLASSES IN CHART
047 VSF,0 / VALUE OF 1 Y AXIS INCREMENT OF THE TTY
048 LOLIM,0/ CHART LOWER LIMIT
049 UPLIM,0/ CHART UPPER LIMIT
050 /
051 TEMP,0/ TEMP STORE
052 /
053 PAUSE
*
```



```

L
001 PAGE
002 /FOLLOWS DECIMAL PRINT SUBROUTINE
003 /THIS SUPRESSES ALL NON SIGNIFICANT ZEROES
004 /
005 DECPRT, 0
006      DCA VALUE
007      TAD (-4
008      DCA CNTIN
009      TAD RESET
010      DCA RTCRKT+1
011      DCA DIGIT
012      TAD CNTRZA
013      DCA CNTRZB      /SET COUNT TO 4
014      TAD ADDRZA
015      DCA ARROW      /SET POINTER
016      SKP
017      DCA VALUE
018      CLL
019      TAD VALUE
020 ARROW, TAD TENPWR
021      SZL
022      ISZ DIGIT
023      SZL
024      JMP ARROW-3
025      CLA
026      TAD DIGIT /GET DIGIT
027      JMS RTCRKT /RIGHT CORRECT FIG.
028      TAD K260 /CONVERT ASCII
029 TTYGO,JMS TYPE
030      DCA DIGIT
031
032      ISZ ARROW /INCR. POINTER
033      ISZ CNTRZB /DONE 4 ?
034      JMP ARROW-1 /NO
035      JMP I DECPRT
036 ADDRZA,TAD TENPWR
037 CNTRZA,-4
038 TENPWR, -1750
039      -0144
040      -0012
041      -0001
042 K260,260
043 VALUE,0
044 DIGIT,0
045 CNTRZB,0
046 RTCRKT, 0
047      ISZ CNTIN
048      JMP .+2
049      JMP I RTCRKT
050      SZA CLA
051      JMP .+3
052      TAD (240
053      JMP TTYGO
054      TAD (7410
055      DCA RTCRKT+1
056      TAD DIGIT
057      JMP I RTCRKT
058 CNTIN,0

```